NAME = "Kisa"
DESCRIPTION = "Kisa -- live spellchecker!"
VERSION = 0.63
AUTHOR = "Pete Black"
EMAIL = "theblackpeter@gmail.com"
LICENSE = "GPL"

TEMPLATE = app

SOURCES += kisa.cpp \
           main.cpp \
           kisawidget.cpp \
           kisathread.cpp \
           kisasettings.cpp

HEADERS += kisa.h \
           kisawidget.h \
           kisathread.h \
           kisasettings.h

CONFIG += warn_on \
          thread \
          qt \
          debug_and_release

QT += dbus

# uncomment to enable full debug output, such as ever event received
# DEFINES += FULL_DEBUG

TARGET = kisa

LIBS += -laspell \
        ../bin/libkisa.a

TARGETDEPS += ../bin/libkisa.a

FORMS += ./ui/kisawidget.ui \
         ./ui/kisasettings.ui

UI_DIR = ./ui
MOC_DIR = ./moc
OBJECTS_DIR = ./obj

DESTDIR = ../bin

QMAKE_CXXFLAGS_DEBUG += -fno-inline

QMAKE_CXXFLAGS_RELEASE += -DQT_NO_DEBUG_OUTPUT

RESOURCES += kisa.qrc

target.path = /bin

doc.path = /share/doc/kisa-$$VERSION
doc.files += ../resources/doc/ChangeLog \
             ../resources/doc/COPYING \
             ../resources/doc/README \
             ../resources/doc/FAQ

icon.path = /share/pixmaps
icon.files += ./icons/kisa.png

desktop.path = /share/applications
desktop.files += ../resources/kisa.desktop

for(lang, LINGUAS) {
  exists(../resources/translations/kisa_$${lang}.qm) {
    translations.files += ../resources/translations/kisa_$${lang}.qm
    translations.path = /share/kisa/translations
  }
}

application_icons.path = /share/kisa/icons
application_icons.files =
ICONS = $$system(ls icons/)
for(icon_file, ICONS) {
  application_icons.files += icons/$$icon_file
}

mappings_override.path =  /share/kisa
mappings_override.files +=  ../resources/mappings_override.txt

INSTALLS += target \
            desktop \
            doc \
            icon \
            translations \
            application_icons \
            mappings_override

isEmpty(translations.files) {
  INSTALLS -= translations
}

TRANSLATIONS += kisa_sv.ts \
                kisa_uk.ts

INCLUDEPATH += lib

MAKEFILE = makefile
